ISTITUTO NAZIONALE FISICA NUCLEARE

INFN/BE - 78/3
23 Gennaio 1978

U. Abbondanno, A. Boiti, F. Demanins and M.R. Malisan:

DUMAN - A FORTRAN PROGRAMME FOR THE AUTOMATICAL ANALYSIS OF GAMMA-RAY AND CHARGED-PARTICLE COMPLEX SPECTRA FROM SOLID STATE DETECTORS

73

SERVIZIO RIPRODUZIONE DELLA
SEZIONE DI TRIESTE DELL'INFN

DUMAN - A FORTRAN PROGRAMME FOR THE AUTOMATICAL ANALYSIS OF GAMMA-RAY AND CHARGED-PARTICLE COMPLEX SPECTRA FROM SOLID STATE DETECTORS

U. Abbondanno, A. Boiti, F. Demanins and M.R. Malisan

Istituto di Fisica dell'Universita', Trieste

Istituto Nazionale di Fisica Nucleare, Sezione di Trieste

1. - INTRODUCTION

In the analysis of spectra of charged-particles or gamma-rays recorded with a solid state detector, the shapes of the peaks often can be assumed to be pure Gaussian functions superimposed to a slowly varying background. In such cases the use of a non-iterative method presents some remarkable advantages with respect to other methods. In fact, it requires a very small computer time if compared to the usual iterative search techniques, a relatively little computer memory, and it doesn't need an initial estimate of the parameters of the peaks, like center and full-width at half-maximum (FWHM).

Non-iterative methods ([1]) have been succesfully applied both to the analysis of gamma ray spectra from NaI(Tl) scintillation detectors ([2]) and to the analysis of charged-particle and gamma-ray spectra from solid state detectors ([3]). Recently, a method of this kind was developed for the automatical analysis of spectra ([4]), in which the second derivative of the spectrum ([5]) is used in order to recognize the presence of a peak, and a non-iterative fitting technique is used to extract the physically interesting parameters.

This analysis is performed by the computer programme DUMAN, written in FORTRAN IV for the HP 2100S computer of the Istituto di Fisica dell'Università di Trieste. In this report we describe the structure of the programme (section 2) and the input data (section 3). The complete listing is given in the Appendix. Since the mathematical apparatus and the flowing of the programme have been presented in detail elsewhere ([4]) for what concerns both the peak finding procedure ([5]) and the fitting technique ([3]), in the present report we shall simply refer to the quantities defined in the above mentioned papers.

2. - DESCRIPTION OF THE PROGRAMME

The programme consists of a main programme and of some subroutines in order to achieve a complete modularity.

Main programme:

DUMAN - It controls the whole flow of the analysis and calls the subroutines involved in the calculations.

Subroutines:

TITLE and DATA - They write the name of the programme, read the input data, report them on teleprinter.

RIDIN - It reads the spectrum from punched paper tape according to the reading technique described in ref.([4]).

COOK - It smooths the spectrum for the determination of the background intervals.

CIGEI and SECDF - They calculate the second difference S_i and the standard deviation F_i of the spectrum, together with the weighting coefficients C_{ij} described in refs.([4]) and ([5]).

HILLS - It identifies the peaks from the analysis of the mutual behaviour of S_i and F_i.

FLAT - It determines the zones in the spectrum through which the background has to be calculated.

SOTFO - It subtracts the calculated background from the spectrum.

TOP - It determines the points to be used in the non-iterative fit.

FIT2R - It performs the non-iterative fit and determines the center and the standard deviation of the peaks.

PARAM - It writes the parameters of the peaks.

BAKL - It subtracts the calculated peaks from the spectrum.

FIT1R - It determines the area of the peaks.

3. - DESCRIPTION OF THE INPUT DATA

In the version of the programme presented in this report, five input cards are necessary for the analysis of a spectrum.

CARD 1 (FORMAT I5) contains KFORM

KFORM is a parameter concerning the FORMAT (see Appendix) of the punched paper tape to be analyzed.

CARD 2 (FORMAT 36A2) contains ITITLE

ITITLE is a label for the spectrum

CARD 3 (FORMAT 16I5) contains LU, NZETA, NPASSO, NSM, I53, NHIL, NSING, NDOUB

LU is the number of the logic unit for a detailed printing of the results (such as the spectrum N_i, the calculated background B_i, the spectrum after background subtraction, the second difference S_i and its standard deviation F_i) which allows one to follow the analysis step by step. LU is usually taken equal to the dummy output, and only the parameters of the analyzed peaks are written in the routine operating of the programme. NZETA is the number of the smoothing iterations in the calculation of the second difference, as explained in ref.([5]).

NPASSO is the smoothing step of the NZETA iterations.

NSM is the smoothing window used in the subroutine COOK for the smoothing of the spectrum in the search of the background zones.

I53 is the minimum distance between the points I_3 and I_5 ([4]) that is acceptable for the identification of a peak.

NHIL is the maximum acceptable distance between I_3 (or I_5) and the nearest channel i for which $S_i > F_i$.

NSING, NDOUB. If a peak is single and the channel nearest to the center is I, the non-iterative fit is performed from the channel (I-NSING) to the channel (I+NSING). If a peak is overlapped to another one, the fit is performed from the channel (I±NSING) (the sign plus or minus is taken according to wether the contaminant peak lies, respectively, to the left or to the right of the considered peak) to the channel (I∓NDOUB).

CARD 4 (FORMAT 16I5) contains LENF, NPART, NARR, LSG, IOVF

LENF is the length of the spectrum

NPART, NARR identify the first and the last channel of the zone of the spectrum that has to be analyzed.
LSG is the length of the segment of the spectrum considered in the reading routine RIDIN.
IOVF is the number of channels of the overlapping zone between two consecutive segments.

CARD 5 (FORMAT 3 (F10.3)) contains ZERO, CONV, FACT
ZERO is the value (in keV) of the energy scale at channel zero.
CONV is the conversion factor of the scale (keV/channel).
FACT is the percentage of the maximum acceptable asymmetry between the positive maxima M_1 and M_2 of the smoothed second difference S_i: if the condition

$$(1-FACT) \leq (M_1/M_2) \leq (1+FACT)$$

is fulfilled, the peak is analyzed as single; in the contrary case, the peak is analyzed as multiple.

After the complete analysis of a spectrum, the programme begins the analysis of another spectrum, by reading CARD 1; a value of KFORM=99 causes the programme to stop.

R E F E R E N C E S

(1) W. Zimmermann, Rev. Sci. Instr. 32, 1063 (1961).

(2) T. Mukoyama, Nucl. Instr. and Meth. 125, 289 (1975).

(3) U. Abbondanno, A. Boiti and F. Demanins, Nucl. Instr. and Meth., 142, 605 (1977)

(4) U. Abbondanno, A. Boiti, F. Demanins and M.R. Malisan, Nucl. Instr. and Meth., in press.

(5) M.A. Mariscotti, Nucl. Instr. and Meth. 50, 309 (1967).

APPENDIX

```
0001  FTN4,L
0002        PROGRAM DUMAN
0003        DIMENSION SMU(560),Q(200)
0004        INTEGER C(45)
0005        COMMON AN(560),S(560),F(560)
0006        COMMON BACK(200),SPUL(200),ESP(200)
0007        COMMON ROW1(200),ROW2(200)
0008        CALL TITLE
0009  306   READ(7,105) KFORM
0010  105   FORMAT (4I5)
0011        IF(KFORM.EQ.99) GO TO 999
0012        CALL DATA(LU,NZETA,NPASSO,LENF,NPART,NARR,LSG,I53,NHIL,
0013       1IOVF,NSING,NDOUB,NSM,ZERO,CONV,FACT,KNOK,NSG,LSS,LUS)
0014        IF(NPART.EQ.1) GO TO 303
0015        NPAR =NPART-1
0016        CALL RIDIN(.TRUE.,KFORM,NPAR,IK)
0017  303   NPE=0
0018        IF (KNOK.EQ.0) GO TO 306
0019        NZONE=0
0020        DO 202 LAV=1,NSG
0021        IF(LAV.EQ.NSG) LSS=LUS
0022        IF(LAV.GT.1) GO TO 304
0023        NOOR=NPART+LSS-1
0024        NPAR = LSS
0025        IK=1
0026        GO TO 305
0027  304   NPART=NOOR-IOVF+1
0028        NOOR=NPART+LSS-1
0029        NPAR =LSS-IOVF
0030        IK=IOVF+1
0031        IF(IK.EQ.1) GO TO 305
0032        DO 204 I=1,IOVF
0033  204   AN(I)=AN(I+LSG)
0034  305   CALL RIDIN(.FALSE.,KFORM,NPAR,IK)
0035        WRITE (6,118) NPART,NOOR
0036  118   FORMAT(5X,"SEGMENT FROM CHANNEL",I5," TO CHANNEL",I5)
0037        WRITE(6,120)
0038  120   FORMAT(5X,"*****************************************",/)
0039        CALL COOK (LSS,NSM,SMU)
0040        NPAR = NPART-1
0041        NLB1=1
0042        NHB1=1
0043        NHB2=1
0044  205   M=NZETA*NPASSO
0045        MPIU=M+1
0046        MOOR=M+2
0047        CALL CIGEI(NZETA,NPASSO,M,C)
0048        CALL SECDF(LSS,0,MOOR,C)
0049  209   CALL HILLS(NHB2,LSS,I53,MI3,MI5,NED,NHIL)
0050        IF(NED.EQ.1) GO TO 202
0051        MI03 = MI3+NPAR
0052        MI05 = MI5+NPAR
0053        CALL FLAT(MI3,MI5,NLB1,NLB2,NHB1,NHB2,LSS,SMOD1,SMOD2,MPIU,IV,SMU)
0054        IF (IV.EQ.0) GO TO 801
0055        NHB2 = MI5+1
0056        GO TO 209
0057  801   NLOB1=NLB1+NPAR
0058        NLOB2=NLB2+NPAR
0059        NLOB3=NHB1+NPAR
0060        NLOB4=NHB2+NPAR
0061        NZONE=NZONE+1
0062        NEP=0
0063        WRITE(6,121) NZONE
0064  121   FORMAT(5X,"ANALYSIS OF THE ZONE NUMBER",I5)
0065        WRITE(6,122)
0066  122   FORMAT(5X,"*********************************",/)
0067        WRITE(6,123) NLOB1,NLOB2
0068  123   FORMAT(5X,"LOW ENERGY BACKGROUND CALCULATED FROM CHANNEL",I6,"  TO
0069       1 CHANNEL",I6)
0070        WRITE(6,124)NLOB3,NLOB4
0071  124   FORMAT(5X,"HIGH ENERGY BACKGROUND CALCULATED FROM CHANNEL",I6,"  T
0072       1O CHANNEL",I6,/)
```

```
0073          MU = NHB2-NLB1+1
0074          MOM1=NLB1-MPIU
0075          IF (MOM1.LT.1) MOM1 = 1
0076          MOM2=NHB2+MPIU
0077          IF (MOM2.GT.LSS) MOM2 = LSS
0078          MOM = MOM2-MOM1+1
0079          IF (NLB1.LE.MPIU) MPIU = NLB1-1
0080          MUM = MU+MPIU
0081          CALL SOTFO(NLB1,NLB2,NHB1,NHB2,SMOD1,SMOD2,MPIU,MOM1,MOM2,MOM)
0082    777   CALL SECDF(MOM,MOM1,MOOR,C)
0083          WRITE(LU,131)
0084    131   FORMAT(1X,"CHANNEL",8X,"AN(I)",8X,"BACK(I)",8X,"SPUL(I)",8X,"S(I)"
0085         1,8X,"F(I)",/)
0086          K = MPIU
0087          DO 234 J=NLB1,NHB2
0088          K=K+1
0089          L=J+NPAR
0090    234   WRITE(LU ,132) (L,AN(J),BACK(K),SPUL(K),S(K),F(K))
0091    132   FORMAT(3X,I5,3X,5(F10.5,5X))
0092          NK = MOOR
0093    409   CALL HILLS(NK,MUM,I53,MI3,MI5,NED,NHIL)
0094          IF(NED.EQ.0) GO TO 410
0095          NHB2=NHB1
0096          GO TO 205
0097    410   CONTINUE
0098          WRITE (LU,105) MI3,MI5
0099          CALL TOP(MI3,MI5,FACT,NSING,NDOUB,I1,I2,ISTAR)
0100          DO 709 I=I1,I2
0101          ROW1(I)=1.
0102          ROW2(I)=I
0103          X =(SPUL(I-1)/SPUL(I+1))
0104    709   ESP(I) = ALOG(X)
0105          CALL FIT2R(I1,I2,AK1,AK2,ROW1,ROW2,ESP)
0106          IF(AK2.GT.0.) GO TO 406
0107          NK = MI5
0108          GO TO 409
0109    406   SIGMA=SQRT(2./AK2)
0110          CENTR =-AK1/AK2
0111          ENF=2.*SIGMA*CONV*1.17741
0112          CONTR = CENTR+NLOB1-1-MPIU
0113          IF (CONTR.GE.FLOAT(NLOB1)) GO TO 756
0114          ISTAR = 0
0115          GO TO 755
0116    756   EN=ZERO+CONV*CONTR
0117          DO 716 I=I1,I2
0118          X=FLOAT(I)
0119    716   ROW1(I)=(EXP(-(X-CENTR)**2/(2*SIGMA**2)))/(2.506628275*SIGMA)
0120          CALL FIT1R(I1,I2,AK1,ROW1,SPUL)
0121          IB1=ISTAR-IFIX(2.3*SIGMA)
0122          IB2=ISTAR+IFIX(2.3*SIGMA)
0123          IF (IB2.GT.MUM+MPIU) GO TO 755
0124          SUMB=0.
0125          DO 751 J1=IB1,IB2
0126    751   SUMB=SUMB+SPUL(J1)
0127          SBAC=0.
0128          DO 752 J2=IB1,IB2
0129    752   SBAC=SBAC+BACK(J2)
0130          FOND=SQRT(SUMB+SBAC)
0131          NPE=NPE+1
0132          NEP=NEP+1
0133          CALL PARAM(NPE,CONTR,EN,ENF,AK1,FOND,SBAC,NEP)
0134    755   CONTINUE
0135          CALL BAKL(MUM,ISTAR,I1,I2,AK1,CENTR,SIGMA,SPUL)
0136          GO TO 777
0137    202   CONTINUE
0138          IF(NOOR.EQ.LENF) GO TO 306
0139          NPAR =LENF-NOOR
0140          CALL RIDIN(.TRUE.,KFORM,NPAR,IK)
0141          GO TO 306
0142    999   WRITE(6,998)
0143    998   FORMAT(20X,"DUMAN IS A FASCINATING BLACK DOG",///)
0144          STOP
0145          END

**  NO ERRORS**     PROGRAM = 02787       COMMON = 05360
```

```
0146          SUBROUTINE TITLE
0147          WRITE(6,101)
0148   101    FORMAT("1",///)
0149          WRITE(6,102)
0150   102    FORMAT(40X,"**********************************")
0151          WRITE(6,103)
0152   103    FORMAT(40X,"*                                *")
0153          WRITE(6,103)
0154          WRITE(6,104)
0155   104    FORMAT(40X,"*   P R O G R A M   D U M A N   *")
0156          WRITE(6,103)
0157          WRITE(6,103)
0158          WRITE(6,102)
0159          WRITE(6,101)
0160          RETURN
0161          END

**  NO ERRORS**     PROGRAM = 00136       COMMON = 00000


0162          SUBROUTINE DATA(LU,NZETA,NPASSO,LENF,NPART,NARR,LSG,I53,NHIL,
0163         1IOVF,NSING,NDOUB,NSM,ZERO,CONV,FACT,KNOK,NSG,LSS,LUS)
0164          DIMENSION ITITLE(36)
0165          DO 201 LS=1,36
0166   201    ITITLE(LS)=2H
0167          READ(7,106)(ITITLE(LS),LS=1,36)
0168   106    FORMAT(36A2)
0169          WRITE(6,107)(ITITLE(LS),LS=1,36)
0170   107    FORMAT(30X,36A2)
0171          WRITE(6,108)
0172   108    FORMAT("1",/)
0173   302    READ(7,105)LU,NZETA,NPASSO,NSM,I53,NHIL,NSING,NDOUB
0174          READ(7,105)LENF,NPART,NARR,LSG,IOVF
0175   105    FORMAT(16I5)
0176          IF (NZETA.GT.7) NZETA = 7
0177          IF (NPASSO.GT.3) NPASSO = 3
0178          NPART=(NPART/8)*8+1
0179          LSG=(LSG/8)*8
0180          NARR=(NARR/8)*8
0181          IOVF=(IOVF/8+1)*8
0182          READ(7,109)        ZERO,CONV,FACT
0183   109    FORMAT(8F10.3)
0184          WRITE(6,110) LENF
0185   110    FORMAT(5X,"SPECTRUM OF ",I5," CHANNELS",/)
0186          WRITE(6,111) LSG
0187   111    FORMAT(5X,"SUBGROUPS OF  ",I5," CHANNELS")
0188          WRITE(6,112) IOVF
0189   112    FORMAT(5X,"OVERLAP BETWEEN THE GROUPS OF",I4," CHANNELS")
0190          WRITE(6,113) NPART,NARR
0191   113    FORMAT(5X,"ANALYSIS FROM CHANNEL ",I5," TO CHANNEL  ",I5,/)
0192          WRITE(6,115) NZETA,NPASSO
0193   115    FORMAT(5X,I1,"  SMOOTHING ITERATIONS WITH A STEP OF ",I1,/)
0194          WRITE(6,116) ZERO,CONV
0195   116    FORMAT(5X,"ENERGY SCALE: ",F10.4," KEV=0 ",F10.4," KEV/CHANNEL")
0196          WRITE (6,118) NSM,I53,NHIL
0197   118    FORMAT (5X,"NSM =",I2," I5MI3 =",I2," NHILLS =",I2,//)
0198          KNOK=NARR-NPART+1
0199          NSG=(KNOK-1)/LSG+1
0200          LSS=LSG+IOVF
0201          IF(LSS.LE.560) GO TO 301
0202          WRITE(6,117)
0203   117    FORMAT("        TOO LONG INTERVAL. RE-ENTER DATA   ")
0204          PAUSE
0205          GO TO 302
0206   301    LUS=KNOK-(NSG-1)*LSG
0207          RETURN
0208          END

**  NO ERRORS**     PROGRAM = 00570       COMMON = 00000
```

```
0209          SUBROUTINE RIDIN(GARB,KFORM,KANAL,IK)
0210          DIMENSION AA(8)
0211          COMMON AN(560),S(560),F(560)
0212          LOGICAL GARB
0213          NOTT = KANAL/8
0214          DO 1 II=1,NOTT
0215          IF (KFORM.EQ.0) GO TO 2
0216          IF (KFORM.EQ.1) GO TO 5
0217          GO TO 3
0218    5     READ (5,9) (AA(I),I=1,8)
0219          GO TO 3
0220    2     READ (5,8) (AA(I),I=1,8)
0221    3     IF (GARB) GO TO 1
0222          DO 4 K=1,8
0223          AN(IK) = AA(K)
0224    4     IK = IK+1
0225    1     CONTINUE
0226    8     FORMAT (7X,8(F6.0,1X))
0227    9     FORMAT (8(F6.0,1X))
0228          RETURN
0229          END

**  NO ERRORS**     PROGRAM = 00163       COMMON = 03360


0230          SUBROUTINE COOK (LSS,NSM,SMU)
0231          DIMENSION SMU(1)
0232          COMMON AN(560)
0233          IF (NSM.EQ.0) RETURN
0234          D = 2*NSM+1
0235          L = LSS-NSM
0236          IN = NSM+1
0237          DO 1 I=IN,L
0238          LI = I-NSM
0239          LS = I+NSM
0240          S = 0.
0241          DO 2 J=LI,LS
0242    2     S = S+AN(J)
0243    1     SMU(I) = S/D
0244          RETURN
0245          END

**  NO ERRORS**     PROGRAM = 00102       COMMON = 01120


0246          SUBROUTINE CIGEI(NZETA,NPASSO,M,C)
0247          INTEGER C(1),B(45),FUC(45)
0248          DO 100 K1 = 1,45
0249    100   C(K1) = 0
0250          C(M+1) = 1
0251          C(M+2) = -2
0252          C(M+3) = 1
0253          IF(NZETA .EQ.0) GO TO 302
0254          NMAX=2*M+3
0255          DO 301 L = 1,NZETA
0256          DO 300 IL = 1,NMAX
0257          K3 = IL-NPASSO
0258          IF (K3.LE.0) K3 = 1
0259          K4 = IL + NPASSO
0260          IF (K4.GT.NMAX) K4 = NMAX
0261          B(IL) = 0
0262          DO 400 KO = K3,K4
0263    400   B(IL) = B(IL) + C(KO)
0264    300   FUC(IL)=B(IL)
0265          DO 99 N = 1,NMAX
0266    99    C(N)=FUC(N)
0267    301   CONTINUE
0268    302   RETURN
0269          END

**  NO ERRORS**     PROGRAM = 00266       COMMON = 00000
```

```
0270          SUBROUTINE SECDF(NEST,KRUK,IO,C)
0271          INTEGER C(1)
0272          COMMON AN(560),S(560),F(560)
0273          COMMON BACK(200),SPUL(200),ESF(200)
0274          DO2 I=1,200
0275          S(I) = 0.
0276    2     F(I) = 0.
0277          DO 12 I=IO,NEST
0278          JO=I-IO+1
0279          JF=I+IO-1
0280          IF (JF.GT.NEST) GO TO 11
0281          SUMS=0.
0282          SUMF=0.
0283          DO 22 L=JO,JF
0284          K=L-I+IO
0285          IF(KRUK.NE.0) GO TO 101
0286          DSUMS= C(K)*AN(L)
0287          DSUMF =   (C(K)**2)*AN(L)
0288          GO TO 20
0289    101   LIF=L+KRUK-1
0290          DSUMS= C(K)*SPUL(L)
0291          DSUMF=(C(K)**2)*AN(LIF)
0292    20    SUMS=SUMS+DSUMS
0293    22    SUMF=SUMF+DSUMF
0294          S(I)=SUMS
0295    12    F(I)=SQRT(SUMF)
0296    11    RETURN
0297          END

**  NO ERRORS**     PROGRAM = 00236       COMMON = 04560


0298          SUBROUTINE HILLS (IP,IA,I5MI3,I3,I5,NED,NHIL)
0299          COMMON AN(560),S(560),F(560)
0300          NED = 0
0301          KU = IA-1
0302          I5 = IP
0303    11    KP = I5+1
0304          IF (KP.GT.KU) GO TO 7
0305          DO 1 I=KP,KU
0306          IF (S(I).LT.-F(I)) GO TO 2
0307    1     CONTINUE
0308    7     NED = 1
0309          RETURN
0310    2     I3 = I
0311          KS = I3+1
0312          DO 3 I=KS,IA
0313          IF (S(I).GE.-F(I)) GO TO 4
0314    3     CONTINUE
0315    4     I5 = I-1
0316          IF (I5-I3.LT.I5MI3) GO TO 11
0317          I = I3
0318    6     I = I-1
0319          IF (I.LT.IP) GO TO 5
0320          IF (S(I).LT.F(I)) GO TO 6
0321          IF (I3-I.GT.NHIL) GO TO 5
0322          RETURN
0323    5     I = I5
0324    8     I = I+1
0325          IF (I.GT.IA) GO TO 7
0326          IF (S(I).LT.F(I)) GO TO 8
0327          IF (I-I5.GT.NHIL) GO TO 11
0328          I6 = I
0329          RETURN
0330          END

**  NO ERRORS**     PROGRAM = 00208       COMMON = 03360
```

```
0331          SUBROUTINE FLAT(MIM3,MIM5,NLB1,NLB2,NHB1,NHB2,LSS,SMOD1,SMOD2,MPIU
0332         1,IV,AN)
0333          DIMENSION AN(1)
0334          NUM=10
0335          NEM=10
0336          IV = 0
0337    10    NUM=NUM-1
0338          NAM=NUM
0339    1     NAM=NAM-1
0340          IF(NAM.GE.4) GO TO 11
0341          IV = 1
0342          GO TO 6
0343    11    NOB1 = MIM3-NAM-4
0344          NOB2=MIM3-4
0345          IF(NOB1.LT.NHB1) GO TO 1
0346    4     NOB1=NOB1-1
0347          NOB2=NOB2-1
0348          IF(NOB1.LT.NHB1) GO TO 1
0349          SMED1=0.
0350          DO 2 IFOT=NOB1,NOB2
0351    2     SMED1=SMED1+AN(IFOT)
0352          SMOD1=SMED1/(NAM+1)
0353          SM = SMOD1+3.
0354          ST = 0.15*ALOG(SM)*SQRT(SM)
0355          SI=SMOD1-ST
0356          SS=SMOD1+ST
0357          DO 3 KFOT=NOB1,NOB2
0358          IF(AN(KFOT).LT.SI.OR.AN(KFOT).GT.SS) GO TO 4
0359    3     CONTINUE
0360          NEM=NEM-1
0361          NOM=NEM
0362    5     NOM=NOM-1
0363          IF(NOM.GT.4) GO TO 12
0364          IV = 1
0365          GO TO 6
0366    12    NOB3=MIM5+4
0367          NOB4=MIM5+4+NOM
0368          IF(NOB4.GT.LSS) GO TO 5
0369    8     NOB3=NOB3+1
0370          NOB4=NOB4+1
0371          IF(NOB4.GT.LSS) GO TO 5
0372          SMED2=0.
0373          DO 7 LFOT=NOB3,NOB4
0374    7     SMED2=SMED2+AN(LFOT)
0375          SMOD2=SMED2/(NOM+1)
0376          SM = SMOD2+3.
0377          ST = 0.15*ALOG(SM)*SQRT(SM)
0378          SI=SMOD2-ST
0379          SS=SMOD2+ST
0380          DO 9 NFOT=NOB3,NOB4
0381          IF(AN(NFOT).LE.SI.OR.AN(NFOT).GE.SS) GO TO 8
0382    9     CONTINUE
0383          IF((NOB4-NOB1).GE.(199-2*MPIU)) GO TO 10
0384          NLB1=NOB1
0385          NLB2=NOB2
0386          NHB1=NOB3
0387          NHB2=NOB4
0388    6     CONTINUE
0389          RETURN
0390          END

**  NO ERRORS**     PROGRAM = 00387       COMMON = 00000


0391          SUBROUTINE SOTFO(L1,L2,N1,N2,S1,S2,MP,M1,M2,MM)
0392          COMMON AN(560),S(560),F(560)
0393          COMMON BACK(200),SPUL(200),ESP(200)
0394          COMMON ROW1(200),ROW2(200)
0395          IX1 = (L1+L2)/2
0396          IX2 = (N1+N2)/2
0397          CA = (S2-S1)/(IX2-IX1)
0398          K = 0
0399          DO 1 J=M1,M2
0400          K = K+1
```

```
0401   1      BACK(K) = S1+CA*(J-IX1)
0402          DO 2 K=1,MM
0403          ROW1(K)=0.
0404          ROW2(K)=0.
0405   2      ESP(K)=0.
0406          IK = L1-MP-1
0407          DO 3 K=L1,L2
0408          I = K-IK
0409          ROW1(I)=1.
0410          ROW2(I)=I
0411   3      ESP(I)=AN(K)
0412          LI = L2+1
0413          LS = N1-1
0414          DO 4 K=LI,LS
0415          I = K-IK
0416          IF (AN(K).GT.BACK(I)) GO TO 4
0417          ROW1(I)=1.
0418          ROW2(I)=I
0419          ESP(I)=AN(K)
0420   4      CONTINUE
0421          DO 5 K=N1,N2
0422          I = K-IK
0423          ROW1(I)=1.
0424          ROW2(I)=I
0425   5      ESP(I)=AN(K)
0426          CALL FIT2R(1,MM,AK1,AK2,ROW1,ROW2,ESP)
0427          K=0
0428          DO 6 J=M1,M2
0429          K=K+1
0430          BACK(K)=AK1+K*AK2
0431          SPUL(K)=AN(J)-BACK(K)
0432          IF(SPUL(K).LE.0.) SPUL(K)=0.5
0433   6      CONTINUE
0434          RETURN
0435          END

**  NO ERRORS**     PROGRAM = 00401       COMMON = 05360


0436          SUBROUTINE TOP(MI3,MI5,FACT,NSING,NDOUB,I1,I2,ISTAR)
0437          COMMON AN(560),S(560),F(560)
0438          COMMON BACK(200),SPUL(200),ESP(200)
0439          DO 701 I=MI3,MI5
0440          IF(S(I+1).GE.S(I)) GO TO 702
0441   701    CONTINUE
0442   702    ISTAR=I
0443          J=ISTAR
0444   704    J=J-1
0445          IF (J.EQ.1) GO TO 703
0446          IF(S(J-1).LE.S(J)) GO TO 703
0447          GO TO 704
0448   703    IMAX1=J
0449          J=ISTAR
0450   705    J=J+1
0451          IF (J.EQ.200) GO TO 706
0452          IF(S(J+1).LE.S(J)) GO TO 706
0453          GO TO 705
0454   706    IMAX2=J
0455          RAP=S(IMAX1)/S(IMAX2)
0456          IF(RAP.GE.(1+FACT).OR.RAP.LE.(1-FACT)) GO TO 707
0457          I1=ISTAR-NSING
0458          I2=ISTAR+NSING
0459          RETURN
0460   707    IF(RAP.GE.(1+FACT)) GO TO 711
0461          I1=ISTAR-NDOUB
0462          I2=ISTAR+NSING
0463          RETURN
0464   711    I1=ISTAR-NSING
0465          I2=ISTAR+NDOUB
0466          RETURN
0467          END

**  NO ERRORS**     PROGRAM = 00220       COMMON = 04560
```

```
0468          SUBROUTINE FIT2R(I1,I2,AK1,AK2,ROW1,ROW2,Q)
0469          DIMENSION ROW1(1),ROW2(1),Q(1)
0470          AK1=0.
0471          AK2=0.
0472          C1=0.
0473          C2=0.
0474          C3=0.
0475          A=0.
0476          B=0.
0477          DO 1 I=I1,I2
0478          C1=C1+ROW1(I)**2
0479          C2=C2+ROW1(I)*ROW2(I)
0480          C3=C3+ROW2(I)**2
0481          A=A+ROW1(I)*Q(I)
0482    1     B=B+ROW2(I)*Q(I)
0483          DELTA=C1*C3-C2**2
0484          DX1=A*C3-B*C2
0485          DX2=B*C1-A*C2
0486          AK1=DX1/DELTA
0487          AK2=DX2/DELTA
0488          RETURN
0489          END

**  NO ERRORS**     PROGRAM = 00223       COMMON = 00000


0490          SUBROUTINE PARAM(NPE,CONTER,EN,ENF,A,FOND,SBAC,NEP)
0491          IF(NEP.EQ.1) WRITE(6,712)
0492    712   FORMAT(7X,"PEAK",2X,"CENTER",4X,"ENERGY",7X,"FWHM",4X,"AREA",6X,"E
0493         1RROR",4X,"BACKGROUND",/)
0494          WRITE(6,713) NPE,CONTER,EN,ENF,A,FOND,SBAC
0495    713   FORMAT(5X,I5,6(F10.2),/)
0496          RETURN
0497          END

**  NO ERRORS**     PROGRAM = 00103       COMMON = 00000


0498          SUBROUTINE BAKL(MUM,ISTAR,I1,I2,A,CENTR,SIGMA,SPUL)
0499          DIMENSION SPUL(1)
0500          IK1=I1+1
0501          IK2=I2-1
0502          IF(ISTAR.EQ.0) GO TO 741
0503          SKUK=SQRT(SPUL(ISTAR))
0504          B = 2*SIGMA**2
0505          C = 2.506628275*SIGMA
0506          DO 720 I=1,MUM
0507          X=FLOAT(I)
0508          X = A*EXP(-(X-CENTR)**2/B)/C
0509          SPUL(I)=SPUL(I)-X
0510          IF(SPUL(I).LT.1.) SPUL(I)=1.
0511    720   CONTINUE
0512          DO 740 KUK=IK1,IK2
0513          IF(SPUL(KUK).LE.SKUK) GO TO 741
0514    740   CONTINUE
0515          RETURN
0516    741   DO 743 KEK=IK1,IK2
0517    743   SPUL(KEK)=1.
0518          RETURN
0519          END

**  NO ERRORS**     PROGRAM = 00192       COMMON = 00000
```

```
0520          SUBROUTINE FIT1R(I1,I2,AK1,ROW1,ESP)
0521          DIMENSION ROW1(1),ESP(1)
0522          AK1=0.
0523          C1=0.
0524          C2=0.
0525          DO 1 I=I1,I2
0526          C1=C1+ROW1(I)*ESP(I)
0527    1     C2=C2+ROW1(I)**2
0528          AK1=C1/C2
0529          RETURN
0530          END

**  NO ERRORS**     PROGRAM = 00082       COMMON = 00000
```